
ABSTRACT

After returning from SAS Global Forum 2016 in Las Vegas, I 
opened my iPhone’s Health App for the first time and was 
surprised to learn it had dutifully tracked the steps I’d taken since 
the day I bought it, including those during SGF.

At SGF 2016, my wanderings were largely confined to the area 
of the Las Vegas Strip shown above. This included my hotel 
(Flamingo, lower left), the convention center (Sands Expo, 
upper right), and the conference hotel (The Venetian, middle).

• SAS XML LIBNAME used to read exported Health App data.

• Photos/captions added with image/text annotation.

• Hourly grid lines (PDT) shown via reference lines/labels.

• Broken-axis feature used to exclude time sleeping.

iPhone Health App step data lacks the precision needed to do a 
cute Family Circus-like take on the data. So, I opted for the next 
best thing to illustrate my not-unlike Billy wandering – an ODS 
Graphics step plot of the 70,000+ steps taken during SGF!

 Family Circus: © 2015 Bill Keane, Inc. Distributed by King Features Syndicate. Reprinted with permission.

• Flat lines represent time not moving; lengthier ones are 
flights, Opening Session, Hands-On Workshops, Keynotes.

• Lengthier, steeper ascents include walks between hotels,  
sightseeing, circulating at Quad & networking events.





* Step plot of steps tracked by iPhone Health App at SAS Global Forum 2016

See idownloadblog.com/2015/06/10/how-to-export-import-health-data/ for how to get step data;

filename health '/folders/myfolders/export.xml'; * XML data provided by iPhone Health App;

filename map '/folders/myfolders/HealthData.map';

libname health xmlv2 xmlmap=map automap=replace;

proc sql; * Grab step data from iPhone XML;

create table steps as

select input(substr(record_startDate,1,19),YMDDTTM19.) as startDT format=datetime26.,

input(substr(record_endDate,1,19),YMDDTTM19.) as endDT format=datetime26., record_value as steps

from health.Record(where=(record_type="HKQuantityTypeIdentifierStepCount"))

where record_startDate between '2016-04-18' and '2016-04-22 05' order by 1;

data totSteps(keep=time totsteps); * Add cumulative steps for each day;

retain totsteps 0;

set steps end=eof; * Reset cumulative steps if "new morning";

if intck('hour',lag1(enddt),startdt)>4 then do; * 4+ hours of inactivity = "new morning";

put "End of Day, Steps Taken=: " totsteps; * Display steps taken during prior "day";

totsteps=0;

end;

time=startdt; * Time and total steps at beginning of period;

output;

totsteps+steps; * Time and total steps at end of period;

time=enddt;

format time datetime19.;

output;

if eof;

put "End of Day, Steps Taken=: " totsteps; * Display steps taken during prior "day";

data anno; * Annotate step plot with headings and photos;

length image label $100. anchor widthunit textcolor function x1space y1space border textweight $20.;

* Headings for each day of conference;

function="text"; y1space='datavalue'; x1space='datavalue';

border='false'; textcolor="GRAY"; anchor="bottomleft";

width=17; widthunit="percent"; textsize=38; textweight="bold"; y1=24250;

x1='18APR2016:05:20:00'dt; label="MONDAY         APRIL 18       27,621 STEPS"; output;

x1='19APR2016:10:00:00'dt; label="TUESDAY        APRIL 19       16,956 STEPS"; output;

x1='20APR2016:09:45:00'dt; label="WEDNESDAY      APRIL 20       15,749 STEPS"; output;

x1='21APR2016:10:00:00'dt; label="THURSDAY       APRIL 21       13,112 STEPS"; output;

* Macro to insert captioned photos taken at SGF 2016;

%macro photo(lbl=, img=, time=, steps=, width=); * Caption, image, time (x), steps (y), width of photo;

function="Image"; * Image annotation parameters;

image="/folders/myfolders/SGF2017Photos/&img"; anchor="bottomleft";

imagescale='fitwidth'; layer='back'; border='true '; linethickness=10; linecolor="black";

drawspace="wallpercent"; x1="&time"dt; y1=&steps; width=&width; widthunit='percent';

x1space='datavalue'; y1space='datavalue';

output;

function="text"; * Caption annotation parameters;

label="&lbl"; anchor="topleft"; border='false'; textcolor="BLACK";

width=20; widthunit="percent"; textsize=14; textweight="bold";

x1space='datavalue'; y1=y1-100; y1space='datavalue';

output;

%mend;

* Day 1 photos;

%photo(lbl=VENETIAN LOBBY, img=VenetianLobby.jpg, time=18APR2016:19:45:00, steps=7000, width=9.41);

%photo(lbl=VENETIAN SHOPS, img=VenetianShopsV2.jpg, time=18APR2016:05:45:00, steps=18250, width=17.42);

%photo(lbl=ORD TO LAS, img=SpiritOrdToLAS.jpg, time=18APR2016:05:45:00, steps=7000, width=10.53);

%photo(lbl=CTA BLUE LINE, img=BlueLine.jpg, time=18APR2016:14:00:00, steps=1000, width=12.41);

%photo(lbl=FLAMINGO ROOM VIEW, img=ViewFlamingoRoom.jpg, time=18APR2016:05:45:00, steps=11600, width=12.30);

%photo(lbl=SGF MINGLE, img=SgfMonEveMingle.jpg, time=18APR2016:22:30:00, steps=15350, width=6.30);

* Day 2 photos;

%photo(lbl=E-MINER HANDS-ON, img=SasEmHandsOn.jpg, time=19APR2016:10:30:00, steps=6750, width=9.38);

%photo(lbl=SAS STUDIO HANDS-ON, img=SasStudioStatHandsOn.jpg, time=19APR2016:17:30:00, steps=1000, width=8.95);

%photo(lbl=WELCOME RECEPTION (QUAD), img=QuadWhatsNewWithSGProcs.jpg, time=19APR2016:10:30:00, steps=19800, width=18.09);

%photo(lbl=OPENING SESSION, img=OpeningSession.jpg, time=19APR2016:10:30:00, steps=11750, width=15.87);

* Day 3 photos;

%photo(lbl=TECHNOLOGY CONNECTION, img=TechConnection.jpg, time=20APR2016:10:15:00, steps=18150, width=21.12);

%photo(lbl=KICKBACK PARTY, img=BrooklynBowlKickback.jpg, time=20APR2016:10:15:00, steps=9100, width=10.45);

%photo(lbl=REDEFINING SUCCESS, img=AriannaHuffington.jpg, time=20APR2016:19:30:00, steps=1000, width=8.50);

%photo(lbl=DATA @FORD, img=Ford.jpg, time=20APR2016:18:15:00, steps=13500, width=6.54);

%photo(lbl=VIYA, img=Viya.jpg, time=20APR2016:13:45:00, steps=1000, width=6.46);

* Day 4 photos;

%photo(lbl=THE SAS PACK, img=SasPack.jpg, time=21APR2016:18:30:00, steps=1000, width=9.65);

%photo(lbl=GTL HANDS-ON, img=GtlHandsOn.jpg, time=21APR2016:10:30:00, steps=14050, width=21.68);

%photo(lbl=EXCEL PLAYGROUND, img=ExcelPlayground.jpg, time=21APR2016:10:30:00, steps=9700, width=9.15);

%photo(lbl=CASINO, img=Slots.jpg, time=21APR2016:10:30:00, steps=20250, width=4.60);

%photo(lbl=B&B BURGER, img=BANDBBURGER.jpg, time=21APR2016:14:25:00, steps=20250, width=7.41);

%photo(lbl=LAS TO ORD, img=LAStoORD.jpg, time=21APR2016:20:15:00, steps=20250, width=7.45);

data _null_; * Generate hourly reference lines + labels;

length reflines labels $ 4096.;

do d='18APR2016:00:00:00'dt to '21APR2016:00:00:00'dt by 60*60*24;

do h=6 to 27; * 6AM-3AM;

t=d+h*60*60; * Step data is in Chicago time (CDT);

reflines=trim(reflines)||" '"||put(t,datetime16.)||"'dt";

l=t-2*60*60; * Convert Chicago to Vegas time (PDT, -2 hours) for labels;

daylabel='    '; * Add day of week to 1st time label of each day;

if l in ('18APR2016:04:00:00'dt '19APR2016:08:00:00'dt '20APR2016:08:00:00'dt '21APR2016:08:00:00'dt)

then daylabel='*'||upcase(put(datepart(l),downame3.));

labels=trim(labels)||" """||compress(put(timepart(l),timeampm5.)||trim(daylabel)||"""");

end;

end;

call symput('reflines', reflines); * Create macro vars with reference lines/labels for plot;

call symput('labels', labels); * Labels are Vegas times (4AM-1AM);

run;

%put &labels; * Sneak a peek;

ods _all_ close; * Plot steps with a (what else?) step plot;

ods listing image_dpi=150 gpath='/folders/myfolders' device=png;

ods graphics on / reset width=40in height=22in imagename="SGF2017steps" imagefmt=png antialias;

proc sgplot data=totsteps sganno=anno nowall; * Step Plot (Time of Day vs. Cumulative Steps Taken);

step  x=time y=totsteps / lineattrs=(thickness=3pt) ;

yaxis display=(nolabel) valueattrs=(size=8.75pt weight=bold) grid gridattrs=(pattern=dot color=black)

values=(0 to 28000 by 1000) valuesformat=comma9. grid;

xaxis display=(nolabel) type=time /* Broken x-axis for sleeping hours (no steps) */

ranges=('18APR2016:05:21:41'dt-'19APR2016:03:15:37'dt

'19APR2016:09:49:46'dt-'20APR2016:00:35:03'dt

'20APR2016:09:40:54'dt-'21APR2016:01:47:23'dt

'21APR2016:09:48:48'dt-'22APR2016:01:55:52'dt)

values=("18APR2016:05:21:41"dt to "22APR2016:01:55:52"dt by "06:00:00"t);

refline &reflines /

lineattrs=(pattern=dot color=black) labelattrs=(size=8.75pt weight=bold) axis=x splitchar='*'

label=(&labels) labelloc=outside labelpos=min;

footnote height=32pt bold "Retracing My SAS® Global Forum 2016 Steps: Visualizing iPhone Health App Step Data With a (What Else?) Step Plot";

